
反応を経時的に測定して得られたデータについてｔ検定の反復あるいは Dunnettの多重比較を行うこと

には問題があるとされる．なぜそれが悪いのか，ではどうしたらよいのかについて SASの具体的プログラ

ムと共に解説する．

Multiple Comparisons, Repeated Measures, Quadrature

1 はじめに

1群の各個体を時間を追って測定したデータ

表 1 1群の経時的測定データ
時点 t0 t1 t2 · · · tk

個体1 y10 y11 y12 · · · y1k

個体2 y20 y21 y22 · · · y2k

...
...

...
...

...
...

個体n yn0 yn1 yn2 · · · ynk

平均 µ0 µ1 µ2 · · · µk

について，各時点の平均を比較したいことがある．たとえば時点 0と各時点との比較

H01 : µ0 = µ1, H02 : µ0 = µ2, . . . , H0p : µ0 = µk (1)

について，対応のある t検定を反復して論文を提出すると「検定の多重性を考慮せよ」というコ

メントがつくことがある．しかし，通常の Dunnettの検定を実施してもよいものであろうか．

また，2群の個体を経時的に測定したデータ

表 2 2群の経時的測定データ
時点 t1 t2 · · · tk

個体1 y11 y12 · · · y1k

群 1
...

...
...

...
...

個体m ym1 ym2 · · · ymk

平均 µ11 µ12 · · · µ1k

個体m+1 ym+11 ym+12 · · · ym+1k

群 2
...

...
...

...
...

個体n yn1 yn2 · · · ynk

平均 µ21 µ22 · · · µ2k

について，各時点毎に対応のない t検定

H01 : µ11 = µ21, H02 : µ12 = µ22, . . . , H0k : µ1k = µ2k (2)

を反復して実施すると，ここにも検定の多重性の問題が現れる．“臨床試験の統計解析に関するガ

イドライン”では「各時点ごとに 2群間での対応のない検定を繰りかえしがちであるが，それでは

第 1種の過誤の確率が増大する．このようなデータの場合には時点毎の比較ではなく，トレンド

やプロフィルを比較するための特別な手法が必要である．」と述べている．事前にトレンドモデル

が想定できる場合には，群間の差を統計モデルのパラメタで表現することは解析法として本質的

ではあるが，事前にはモデルを想定できないこともあるだろう．

本稿では，時点間の相関を考慮しつつ群平均の多重比較を行う方法について考察する．

2 反復測定時点間の多重比較

2.1 独立誤差の場合

表 1,式 1で示したような反復測定での時点間の平均を比較することを考えよう．次のデータを
例題とする．

data repeated;

input id @;

do time=0 to 3;

input y @;

output;

end;

cards;

301 2.07 1.58 1.84 2.06

305 1.86 2.09 1.76 2.43

306 2.84 2.13 1.62 1.62

307 3.30 2.03 2.88 2.60

308 1.73 1.67 1.15 1.21

309 3.40 2.16 3.20 2.24

310 2.79 1.82 2.38 1.29

311 2.73 3.38 1.93 2.61

;

通常の一元配置モデルでの Dunnettの多重比較を PROC GLMで実行する．

proc glm data=repeated;

class time;

model y = time;

lsmeans time / adjust=dunnett pdiff=control(’0’);

run;

出力 1.1 一元配置の Dunnett比較

General Linear Models Procedure

Least Squares Means

Adjustment for multiple comparisons: Dunnett

TIME Y Pr > |T| H0:

LSMEAN LSMEAN=CONTROL

0 2.59000000

1 2.10750000 0.2868

2 2.09500000 0.2683

3 2.00750000 0.1624

反復測定データを一元配置の方法で解析することは，時点間の相関をゼロと仮定することと同じ
である．PROC MIXEDによる反復測定解析で TYPE=VCの指定を行うと同様の解析が再現で
きる．

proc mixed data=repeated;

class time;

model y = time / ddfm=residual;

repeated time / type=VC subject=id rcorr;

lsmeans time / adjust=dunnett pdiff=control(’0’);

run;

出力 1.2 独立誤差の相関構造

R Correlation Matrix for Subject 1

Row COL1 COL2 COL3 COL4

1 1.00000000

2 1.00000000

3 1.00000000

4 1.00000000

出力 1.3 独立誤差を仮定した Dunnett比較

Differences of Least Squares Means

Effect TIME _TIME Difference Std Error DF t Pr>|t| Adjustment Adj P

TIME 1 0 -0.48250000 0.30616143 28 -1.58 0.1263 Dunnett 0.2868

TIME 2 0 -0.49500000 0.30616143 28 -1.62 0.1171 Dunnett 0.2683

TIME 3 0 -0.58250000 0.30616143 28 -1.90 0.0674 Dunnett 0.1624

37



出力 1.2中，ブランクの相関係数はゼロであることを表している．すべての時点対間の相関係数

をゼロとした反復測定の多重比較は，通常の一元配置多重比較と全く同じ結果をもたらすことが

わかる．

2.2 球面誤差の場合

誤差分散が等しく時点間の共分散もすべて等しいとき，その共分散行列は複合対称であるとい

う．このとき，通常のF検定は妥当である．さらに一般的には，Huyhn and Feldt(1970)が示した

球面性を満たしているときも F検定は妥当である．誤差が球面であるとき，Tukey法や Dunnett

法のような通常の多重比較法も妥当である．
PROC GLMでこのような多重比較を実行するには，個体差をモデルに入れたプログラムを指

定する．

proc glm data=repeated;

class id time;

model y = id time;

lsmeans time / adjust=dunnett pdiff=control(’0’);

run;

出力 2.1 個人差をモデルに入れた Dunnett比較

General Linear Models Procedure

Least Squares Means

Adjustment for multiple comparisons: Dunnett

TIME Y Pr > |T| H0:

LSMEAN LSMEAN=CONTROL

0 2.59000000

1 2.10750000 0.1189

2 2.09500000 0.1073

3 2.00750000 0.0502

PROC MIXED を使って，球面形の誤差構造を指定しても同じ結果が得られる．

proc mixed data=repeated;

class time;

model y = time;

repeated time / type=HF subject=id rcorr;

lsmeans time / adjust=dunnett pdiff=control(’0’);

run;

出力 2.2 球面性を仮定した相関構造

R Correlation Matrix for Subject 1

Row COL1 COL2 COL3 COL4

1 1.00000000 0.44604378 0.49269359 0.51790125

2 0.44604378 1.00000000 0.34194834 0.39242273

3 0.49269359 0.34194834 1.00000000 0.45236522

4 0.51790125 0.39242273 0.45236522 1.00000000

出力 2.3 球面性を仮定した Dunnett比較

Differences of Least Squares Means

Effect TIME _TIME Difference Std Error DF t Pr>|t| Adjustment Adj P

TIME 1 0 -0.48250000 0.23037494 21 -2.09 0.0485 Dunnett-Hsu 0.1189

TIME 2 0 -0.49500000 0.23037494 21 -2.15 0.0435 Dunnett-Hsu 0.1073

TIME 3 0 -0.58250000 0.23037494 21 -2.53 0.0195 Dunnett-Hsu 0.0502

個人差をモデルに入れた検定と誤差に球面性の仮定をおいた検定とは全く同じ結果をもたらす．す

なわち，誤差に球面性が仮定できるなら個人差をモデルに入れた多重比較は妥当である．一元配

置型 Dunnett検定で TIME=0と TIME=3の比較の p値が 0.1624であったのに対し，球面誤差

の場合は 0.0502になっており，差をより鋭敏に検出できることがわかる．反復測定型のデータに

一元配置型の多重比較を施すのが不適切なのは，(相関がほとんどゼロの場合を除き)検出力で損

をするからである．

2.3 非球面誤差の場合

球面型の誤差構造が不適切な場合には，適当な誤差構造を仮定して最尤推定するか，あるいは

標本からそのまま推定された相関構造を元に検定を構成することになる．ところが，特定の相関

構造以外の積分計算は今日のコンピュータをもってしても難しい．そこで，いくつかの近似法が

とられる．

1. 各相関係数の平均を計算し，共通の相関係数とする．

2. Hsu(1992)の因子分析的方法を用いて近似的な積分を行う．

3. 標本相関係数を再現するような多変量 t分布を乱数で再現しその最大値に対する調整 p値を

求める．

Release 6.10以降の PROC GLMでは，LSMEANSステートメント中 ADJUST=DUNNETT
オプションを指定すると，必要な時は自動的に Hsuの近似が行なわれる．たとえば，AR(1)の誤
差構造を想定して Hsu型の近似計算を行うためには次のようにする．

proc mixed data=repeated;

class time;

model y = time;

repeated time / type=AR(1) subject=id rcorr;

lsmeans time / adjust=dunnett pdiff=control(’0’);

run;

図 3.1 AR(1)を仮定した誤差構造

R Correlation Matrix for Subject 1

Row COL1 COL2 COL3 COL4

1 1.00000000 0.27679862 0.07661748 0.02120761

2 0.27679862 1.00000000 0.27679862 0.07661748

3 0.07661748 0.27679862 1.00000000 0.27679862

4 0.02120761 0.07661748 0.27679862 1.00000000

図 3.2 AR(1)を仮定し Hsu型の近似を行った Dunnett比較

Differences of Least Squares Means

Effect TIME _TIME Difference Std Error DF t Pr>|t| Adjustment Adj P

TIME 1 0 -0.48250000 0.26019292 21 -1.85 0.0778 Dunnett-Hsu 0.1783

TIME 2 0 -0.49500000 0.29400632 21 -1.68 0.1071 Dunnett-Hsu 0.2385

TIME 3 0 -0.58250000 0.30269911 21 -1.92 0.0680 Dunnett-Hsu 0.1575

特定の相関構造を仮定しないで，シミュレーションにより調整 p値を求めるには，次のように
する．

proc mixed data=repeated;

class time;

model y = time;

repeated time / type=UN subject=id rcorr;

lsmeans time / adjust=SIMULATE pdiff=control(’0’);

run;

図 4.1 無構成の誤差構造

R Correlation Matrix for Subject 1

Row COL1 COL2 COL3 COL4

1 1.00000000 0.31304700 0.84507324 0.31850774

2 0.31304700 1.00000000 0.09341769 0.56166162

3 0.84507324 0.09341769 1.00000000 0.44463342

4 0.31850774 0.56166162 0.44463342 1.00000000

図 4.2 無構成誤差構造，シミュレーションによる Dunnett比較

Differences of Least Squares Means

Effect TIME _TIME Difference Std Error DF t Pr>|t| Adjustment Adj P

TIME 1 0 -0.48250000 0.24828519 7 -1.94 0.0931 Simulate 0.2058

TIME 2 0 -0.49500000 0.13038405 7 -3.80 0.0067 Simulate 0.0162

TIME 3 0 -0.58250000 0.24890295 7 -2.34 0.0518 Simulate 0.1173

通常の臨床試験や前臨床試験でのデータでは，ランダムな誤差が後の誤差に与える影響よりも

個体差の影響の方がずっと強いと思われる．また，少ない標本から推定された相関係数を真の値

として検定を構成するのは不安定である．ゆえに，TYPE=AR(1)とか TYPE=UNのような相関

構造が適切な例は少なく，球面性 (あるいは複合対称)を仮定した解析が良いことが多いだろうと

いうのが著者の見解である．

3 いわゆる輪切りの検定

3.1 定式化と方法

表 2と式 2に表したように，2つの処置群について k時点測定した試験について考える．岸本

(1997)が示したように，時点間の相関構造に複合対称を仮定すれば，時点毎に行う対応のない t検

定について多重性を調整した p値を得ることができる．その概略は次の通りである．

観測値 Yihrについて，各セルの平均をパラメタとしたモデルを想定する．

Yihr = µih + εihr, i = 1, 2, h = 1, ..., k, r = 1, ..., nih, (3)

同一群の観測値 Yihrは等分散で時点間で等相関と仮定し，さらに等サンプルサイズとする．パラ

メタ推定値 bµ = (bµ11, . . . , bµ1k, bµ21, . . . , bµ2k) 間の分散共分散は，k=3のとき次のように表される．

Var(bµ) = σ2



1 ρ ρ 0 0 0

ρ 1 ρ 0 0 0

ρ ρ 1 0 0 0

0 0 0 1 ρ ρ

0 0 0 ρ 1 ρ

0 0 0 ρ ρ 1


(4)

仮説のコントラスト行列を次のように設定する。

L =


1 0 0 −1 0 0

0 1 0 0 −1 0

0 0 1 0 0 −1

 (5)

コントラスト間の分散共分散は次のようになる。

Var(Lbµ) = 2σ2


1 ρ ρ

ρ 1 ρ

ρ ρ 1

 (6)

よく知られているように，このような等相関の多変量 t分布の積分計算は実行可能であり，これ

により各コントラストの最大値について確率評価することができる．SASでは，PROBMC関数

で Dunnett型の指定をすることにより求められる．

表 3 5%臨界値の表
k=2 ρ

ν 0 0.3 0.5 0.7 0.9 1

4 3.38 3.36 3.31 3.23 3.06 2.78

8 2.72 2.70 2.67 2.62 2.51 2.31

12 2.54 2.53 2.50 2.46 2.36 2.18

16 2.46 2.45 2.42 2.38 2.29 2.12

∞ 2.24 2.23 2.21 2.18 2.11 1.96

k=3

ν 0 0.3 0.5 0.7 0.9 1

6 3.19 3.16 3.10 2.99 2.80 2.45

12 2.75 2.73 2.68 2.61 2.46 2.18

18 2.62 2.60 2.56 2.50 2.36 2.10

24 2.56 2.54 2.51 2.44 2.31 2.06

∞ 2.39 2.38 2.35 2.30 2.19 1.96

k=4

8 3.13 3.09 3.02 2.91 2.69 2.31

16 2.78 2.76 2.71 2.62 2.44 2.12

24 2.68 2.66 2.61 2.53 2.37 2.06

32 2.63 2.61 2.57 2.49 2.34 2.04

∞ 2.49 2.47 2.44 2.38 2.24 1.96

ρ = 0のときは Studentized Maximum Modulus に、ρ = 1のときは多重性を考慮しない場合に対

応する。相関の大きさによる臨界値の変化は小さく，相関が相当大きくないと相関を考慮しない

場合に比べて有利にならないことがわかる。実際の解析ではデータから推定した相関で ρを置き

換えるが，その際の推定誤差は問題にならない．また，多少の不等相関があっても影響ないこと

が示唆される．

3.2 輪切り検定の例題

多重性を考慮した輪切り検定の SASによる実施例を紹介する．データは遅効性の薬剤に関する
もので，投薬後 5時間～8時間のあたりで実薬群と溶媒群とに効果の差があることが期待された
が，いつ効果があるかは事前には明らかになっていなかった．測定した 4時点のうちいずれかで
有意な差があれば効果ありと判定したいが，相関のある 4時点で測定していることの多重性を考
慮しなければならない．

data B;

input group $ id $ @;

do time=5 to 8;

input y @;

output;

end;

cards;

Vehicle 001 3.25 3.55 3.44 1.78

Vehicle 003 2.59 2.80 4.18 2.63

Vehicle 004 3.09 3.29 2.05 1.28

Vehicle 007 2.66 2.86 2.27 1.82

Vehicle 008 2.84 2.43 2.36 2.30

Vehicle 009 2.01 2.86 1.84 2.50

Vehicle 010 3.22 3.36 2.80 2.26

Vehicle 011 3.16 3.85 3.42 1.93

D+P(H) 301 2.07 1.58 1.84 2.06

D+P(H) 305 1.86 2.09 1.76 2.43

D+P(H) 306 2.84 2.13 1.62 1.62

D+P(H) 307 3.30 2.03 2.88 2.60

D+P(H) 308 1.73 1.67 1.15 1.21

42



D+P(H) 309 3.40 2.16 3.20 2.24

D+P(H) 310 2.79 1.82 2.38 1.29

D+P(H) 311 2.73 3.38 1.93 2.61

;

proc mixed data=B;

class group time id;

model y = group(time) / ddfm=residual;

repeated time / type=cs subject=id RCORR;

lsmeans group(time) / corr slice=time;

run;

図 5.1 各セルの最小二乗平均

Least Squares Means

Level LSMEAN Std Error DDF T Pr > |T|

GROUP(TIME) D+P(H) 5 2.59000000 0.20761292 56 12.48 0.0001

GROUP(TIME) Vehicle 5 2.85250000 0.20761292 56 13.74 0.0001

GROUP(TIME) D+P(H) 6 2.10750000 0.20761292 56 10.15 0.0001

GROUP(TIME) Vehicle 6 3.12500000 0.20761292 56 15.05 0.0001

GROUP(TIME) D+P(H) 7 2.09500000 0.20761292 56 10.09 0.0001

GROUP(TIME) Vehicle 7 2.79500000 0.20761292 56 13.46 0.0001

GROUP(TIME) D+P(H) 8 2.00750000 0.20761292 56 9.67 0.0001

GROUP(TIME) Vehicle 8 2.06250000 0.20761292 56 9.93 0.0001

図 5.2 各セル間の相関係数

CORR1 CORR2 CORR3 CORR4 CORR5 CORR6 CORR7 CORR8

1.00 -0.00 0.30 0.00 0.30 0.00 0.30 0.00

-0.00 1.00 -0.00 0.30 0.00 0.30 -0.00 0.30

0.30 -0.00 1.00 0.00 0.30 0.00 0.30 0.00

0.00 0.30 0.00 1.00 0.00 0.30 0.00 0.30

0.30 0.00 0.30 0.00 1.00 0.00 0.30 0.00

0.00 0.30 0.00 0.30 0.00 1.00 0.00 0.30

0.30 -0.00 0.30 0.00 0.30 0.00 1.00 0.00

0.00 0.30 0.00 0.30 0.00 0.30 0.00 1.00

図 5.3 各時点での多重性を調整しない検定

Tests of Effect Slices

Effect Slice NDF DDF F Pr > F

GROUP(TIME) TIME 5 1 56 0.80 0.3751

GROUP(TIME) TIME 6 1 56 12.01 0.0010

GROUP(TIME) TIME 7 1 56 5.68 0.0205

GROUP(TIME) TIME 8 1 56 0.04 0.8521

TIME=6 のとき F=12.01, p=0.0010 で最も効果が大きい．

得られた相関と F値 (t値に変換)を基に，多重性を調整した p値を求める．

data mult_t;

k = 4;

t = sqrt(12.01);

rho = 0.30;

nu = 56;

array lambda{4};

do i=1 to k;

lambda{i} = sqrt(rho);

end;

p = 1 - probmc("DUNNETT2", t, ., nu, k, of lambda1-lambda4);

put p= 6.4;

run;

P=0.0040

TIME=6 での生の p値が 0.0010であったのに対し，多重性を調整した p値は 0.0040になっ

た．今の例では相関が 0.30と小さかったことにより，Bonferroni調整と同じ結果となった．Frison

and Pocock(1992) が 10の臨床試験を調べた報告によれば，visit間の相関は 0.6—0.8程度である

ことが多かった，そのような試験であれば Bonferroni調整よりは精度よく検定できる．

4 おわりに

経時データの多重比較法は，従来そのための方法がわからないという理由で十分に行われてこ

なかったように思われる．原理は意外と簡単であり，SASでの実行も容易であることが示された．

